
__________________________________________________________________________________

KALMAN (BPRIOR=prior vector, 
BTRANS=matrix of coefficients in transition equation, 

EMEAS, ETRANS, PRINT, SILENT, SMOOTH, TERSE,
VBPRIOR=variance of prior, VMEAS=variance factor in measurement equation,

VTRANS=variance factor in transition equation,
XFIXED=X matrix for measurement equation)

list of dependent variables  [| list of independent variables];

__________________________________________________________________________________

Function:

KALMAN estimates linear models using the Kalman filter method.  It can handle fairly general State Space models, but
it is typically used to estimate regression-type models where the coefficients follow a random process over time.

Usage:

The Kalman filter model consists of two parts (the state space form):

     yt = Xt  $t + ,t,        ,t ~ N(0,F2H)  (measurement equation)
       (n×1)  (n×m)(m×1)                        (n×n)

     $t = T*$t-1 + 0t,   0t ~ N(0,F2Q)  (transition equation)
                  (m×m)                           (m×m)

     $0 ~ N ($0, F
2P0)                    ("initial conditions")

The matrices T, H, and Q are assumed to be known, and they each default to the identity matrix if they are not supplied
by the user in the KALMAN options list.  Note that they are not allowed to vary over time, but this constraint can be
easily relaxed by running KALMAN within a loop over the sample.  The NOEMEAS and NOETRANS options are used
to zero the variances of the measurement and transition equations respectively.

The yt and Xt variables are the dependent and independent variables, just as in ordinary least squares.  If you want to use
more than one dependent variable, list all the y variables first, then a |, and then list the X variables for each y (duplicate
the X list if they are the same for every y).  You may want to insert zeros along with the X variables to prevent cross-
equation restrictions (see the Examples).  If Xt is fixed over the sample, use the XFIXED option.

To get a time-varying parameter model, specify VTRANS=Q (the noise-to-signal ratio) and BTRANS=T (if it is not the
identity matrix).

To evaluate the likelihood function for general ARMA(p,q) models, fill the BTRANS and VTRANS matrices with the
estimated coefficients for the model; see Harvey, p.103 for the general form.

Options:

BPRIOR= vector of prior coefficients $0 for measurement equation.  Required if XFIXED is used; otherwise it will be
calculated by default from a regression in the initial observations of the sample.  If the first M observations are not
sufficient to identify the prior, one observation is added and BPRIOR is estimated again.

BTRANS= T = matrix of coefficients in transition equation (default identity matrix).

EMEAS/NOEMEAS indicates the presence of an error term in the measurement equation (NOEMEAS or NOEM is the
same as VMEAS = zero).

ETRANS/NOETRANS indicates the presence of an error term in the transition equation.

PRINT/NOPRINT prints the prior, @STATE, @RES1, @RECRES, @SMOOTH, @RESD.

SILENT/NOSILENT turns all of the printed output.  See also TERSE.

SMOOTH/NOSMOOTH computes fixed-interval smoothed estimates of the state vector $t (stored in @SMOOTH) and
the direct residuals @RESD.

TERSE/NOTERSE turns off most of the output.  See also SILENT.

VBPRIOR= P0 = variance of prior (symmetric matrix).  Required if BPRIOR is specified.  Note:  F2 is factored out of
this matrix.

VMEAS= H = variance factor in measurement equation (symmetric matrix).  Default: identity matrix.  In Harvey’s
notation, this is SHS’.

VTRANS= Q = variance factor in transition equation (symmetric matrix).  Default: identity matrix.  Specifies the
"noise-to-signal ratio" if H = identity matrix.  In Harvey’s notation, this is RQR’.

XFIXED= X matrix for measurement equation, when it is fixed over time.

Examples:

One of the simplest Kalman filter models is equivalent to OLSQ (using a transition equation of $t = $t-1 = $).  This can
be estimated with the command: 

KALMAN(NOETRANS) Y C X;

which produces the same coefficient estimates as OLSQ Y C X ;, but calculates them recursively, along with the
recursive residuals.   

To estimate a Cooley-Prescott "adaptive regression" model where $t follows random walk with a nondiagonal variance
matrix:

KALMAN(VTRANS=NSRMAT) Y C X1 X2;

A "stochastically convergent parameter" model (convergent towards zero in this case, since the transition matrix has roots
less than one):

MFORM(TYPE=DIAG,NROW=3) TMAT=.9;
 KALMAN(BTRANS=TMAT,VTRANS=NSRMAT) Y C X1 X2;

Two dependent variables; note the two lists of exogenous variables, which must be of the same length.  In this case, both
equations are forced to have the same two coefficients.

KALMAN(NOET) Y1 Y2 | C1 X1, C2 X2;

Two dependent variables; in this case, the equations have separate coefficients by building zero variables into the system.

The specification is still somewhat unrealistic because H is identity (same variance and no correlation between errors
in the equations):

ZERO = 0;
KALMAN(NOET) Y1 Y2 | C1 X1 ZERO ZERO, ZERO ZERO C2 X2;

Harvey’s Example 2 (p.116-117) ("signal plus noise" or Cooley-Prescott):

READ Y;  4.4 4.0 3.5 4.6;
SET Q = 4;  SET A0 = 4;  SET P0 = 12;
KALMAN(VT=Q,BPRIOR=A0,VBPRIOR=P0) Y C;

Harvey’s Exercise 1 (p.119) (stochastically convergent):

READ Y;  4.4 4.0 3.5 4.6;
Y4 = Y-4;
SET RHO=.5;  SET Q = 4;  SET A0 = .2;  SET P0 = 3;
KALMAN(BT=RHO,VT=Q,BP=A0,VBP=P0) Y4 C;

Bootstrapping a variance for the transition equation:

SMPL 1,100;
KALMAN(NOETRANS) Y C X1 X2;
UNMAKE @STATE B1-B3;
SMPL 4,100;  DOT 1-3;  D.=B.-B.(-1);  ENDDOT;
COVA D1-D3;  MAT VTOS=@COVA/@S2;
KALMAN (VTRANS=VTOS) Y C X1 X2;

Hyperparameter estimation with ML PROC.  This example estimates the variances of the transition matrix Q, using the
ML PROC.

MFORM(NROW=2,TYPE=SYM) Q;
PARAM Q11,1 Q22,2;
ML  KFQ  Q11,Q22;
? Evaluate log likelihood for ML PROC
PROC KFQ;   
  IF (Q11<=0 .OR. Q22<=0); THEN;   ? Check constraints
    SET @LOGL=@MISS; ? Are variances >0?
  ELSE; DO;
    SET Q(1,1) = Q11; SET Q(2,2) = Q22;
    SUPRES @COEF;
    KALMAN(SILENT,VTRANS=Q) Y C X;
    NOSUPRES @COEF;
  ENDDO;
ENDPROC;

Output:

A standard table of coefficients and standard errors is printed for the final state vector, along with the log likelihood.
The following items are stored (and may be printed):

Name Length Description

@COEF m Final state vector

@SES m Standard errors

@T m T-statistics

@VCOV m * m Variance-covariance matrix

@LOGL 1 Log of likelihood function

@SSR 1 Sum of squared recursive residuals

@S2 1 Variance of recursive residuals

@RES1 #obs * n Prediction errors (one step ahead)

@RECRES #obs * n Recursive residuals (i.i.d.)

@STATE #obs * m Evolving state vectors

@RESD #obs * n Direct residuals (if SMOOTH is on)

@SMOOTH #obs * m Smoothed state vectors

Note that the first few rows in the residuals or state vectors may be zero if those observations were used to calculate
priors.  Also, recursive residuals for OLS regressions can be obtained with REGOPT(CALC) RECRES; and OLSQ.

Method:

The Kalman filter recursively updates the estimate of $t (and its variance), using the new information in yt and Xt for each
observation, so it can be viewed as a Bayesian method.  However, the user does not have to supply priors; they can be
calculated automatically for regression-type models from the initial observations of the sample.  See the Harvey reference
for the actual updating formulas.  If the default prior is singular, KALMAN adds one more observation to the calculation,
unless NOETRAN is being used (it can handle a singular prior).  The smoothed state vectors (if requested) are estimates
based on the full sample; again, see Harvey for the details.

The method of estimation for each time period is maximum likelihood conditional on the data observed to that point.
An orthonormalizing transformation of X is used to improve accuracy.

The F2 factor and the log likelihood are computed from the recursive residuals.  The recursive residuals are et =
CHOL(Ft)

-1vt in Harvey’s notation, so that E[et’et] = I.  If you do not factor F2 out of H, Q, and P0, the estimated @S2
should be close to 1.  If m>1 and the prediction errors are collinear, there may be problems with the resulting F2, standard
errors, recursive residuals, and log likelihood.

